Problem:
Longest Increasing Subsequence

For a given sequence A = {a0, a1, ... , an-1}, find the length of the longest increasing subsequnece (LIS) in A.

An increasing subsequence of A is defined by a subsequence {ai0, ai1, ... , aik} where 0 ≤ i0 < i1 < ... < ik < n and ai0 < ai1 < ... < aik.

Input

n
a0
a1
:
an-1

In the first line, an integer n is given. In the next n lines, elements of A are given.

Output

The length of the longest increasing subsequence of A.

Constraints

1 ≤ n ≤ 100000

0 ≤ ai ≤ 109

Sample Input 1

5
5
1
3
2
4

Sample Output 1

3

Sample Input 2

3
1
1
1

Sample Output 2

1

Solution in Haskell (judge verdict: Accepted)

{-# OPTIONS_GHC -O2 -funbox-strict-fields #-}

import           Control.Applicative
import qualified Data.ByteString.Char8 as B
import           Data.Char
import qualified Data.IntSet           as IS
import qualified Data.Vector.Unboxed   as U

main :: IO ()
main = do
    n <- readLn :: IO Int
    xs <- U.unfoldrN n (B.readInt.B.dropWhile isSpace) <$> B.getContents
    print $ solve xs

solve :: U.Vector Int -> Int
solve xs = IS.size $ U.foldl' step IS.empty xs
  where
    step s x = case IS.lookupGE x s of
        Just y | x < y     -> IS.insert x $ IS.delete y s
               | otherwise -> s
        Nothing            -> IS.insert x s